
Abstract Aggregates are among the most important

linguistic extensions of Answer Set Programming

(ASP), allowing for compact representations of prop-

erties and inductive definitions involving sets of propo-

sitions. Common use cases of aggregates in ASP are

reported in this paper, which mainly focus on the se-

mantics implemented by mainstream solvers, namely

the F-stable model semantics. Other well-established

semantics are also briefly discussed, providing a his-

torical perspective on the foundation of logic programs

with aggregates.

Keywords answer set programming · recursive

aggregates · nonmonotonic reasoning

1 Introduction

ASP is a declarative language for knowledge represen-

tation and reasoning [16], where highly complex compu-

tational problems are modeled by means of logic pro-

grams comprising disjunctive logic rules, possibly us-

ing default negation under stable model semantics in

their bodies [24,25]. One of the origins of ASP is Dat-

alog, a database query language based on logic pro-

gramming. Since constructs for aggregate functions like

sum or count are well-established in database query

languages, it was natural to introduce them into Data-

log [41], and consequently ASP as well [12,19,21,26,

31,34,39]. Indeed, aggregation functions allow for ex-

pressing properties on sets of atoms declaratively, and

have found applications in several contexts. Common

use cases are enforcing functional dependencies, con-

straining nondeterministic guesses, and inductive def-

initions involving recursive aggregations. Some exam-

ples are provided later on in Section 3.

Mainstream ASP solvers [2,3,17,18,20,23,27,33,35]

mostly agree on the semantics of aggregates [19,21],

here referred to as the F-stable model semantics. Some

alternatives will be briefly reviewed in Section 4, but

it is important to stress here that several semantics

agree on a large class of commonly used aggregates,

referred to as convex in the literature [32]; the con-

sensus is lost for non-convex aggregations, which also

introduce a new complexity source for several reasoning

tasks.

According to the F-stable model semantics, a model

I of a propositional program Π is stable if there is no

J ⊂ I satisfying all rules of the program reduct ΠI ,

obtained from Π by removing all rules whose body is

false with respect to I, and by fixing the interpretation

of negated literals in the remaining rules. The seman-

tics of programs with symbolic variables can be defined

in terms of their grounding, obtained by substituting

variables occurring in rules with ground terms in all

possible ways. A few additional insights into the syntax

and semantics of aggregates is given in Section 2.

2 Syntax and Semantics

The notion of F-stable models, as given in the intro-

duction, is not significantly different from the original

definition of stable models for programs without aggre-

gates [25]. Actually, the aspects that require clarifica-

tion are how aggregates are written, how aggregates are

grounded, and how ground aggregates are interpreted.

This section provides sufficient notions to understand

these concepts; familiarity with standard syntax and

semantics of ASP is assumed.

An aggregate element has the following form:

t1, . . . , tm : ϕ

where t1, . . . , tm (m ≥ 1) are terms, and ϕ is a con-

junction of literals; all variables in t1, . . . , tm occur in

ϕ. The intuitive meaning is the set of tuples t1, . . . , tm
for which ϕ holds. An aggregate has the following form:

f{e1; · · · ; en} � t

where f is an aggregation function, e1, . . . , en (n ≥ 1)

are aggregate elements, � is an arithmetic comparator,

and t is a ground term. The intuition is to apply f on

the multiset of the first terms of all tuples represented

by the entries and compare the result to t using �.

Common aggregation functions are #count, #sum, #min,

and #max.

The grounding of programs with aggregates is based

on the notions of global and local variables. Global vari-

ables of a rule are variables occurring in at least one

literal not involved in any aggregation; other variables

are local to the aggregate element they occur in. A

ground instance of a rule is obtained by replacing global
variables with ground terms (of a fixed set), and then

by expanding local variables of each aggregate element.

Specifically, the expansion of an aggregate element com-

prises all ground aggregate elements obtained by substi-

tuting local variables with ground terms in all possible

ways.

Example 1 Consider the following rule:

:- p(X), #sum{Y : q(X,Y,Z)} >= 2,

#sum{Y,Zp : q(X,Y,Zp)} <= 3.

Here, X is a global variable, while Y, Z and Zp are local

variables. Note that the same variable Y is used as a

local variable by the two aggregates above, but this

does not create any link between the two aggregates;

in fact, Y could be replaced by Yp in either of the two

aggregates, and the grounding of the above rule would

not change for any set of ground terms.

The grounding of the rule above, for the set {1, 2}
of ground terms, yields the following ground rules:

:- p(1), #sum{1 : q(1,1,1); 1 : q(1,1,2);

2 : q(1,2,1); 2 : q(1,2,2)} >= 2,

#sum{1,1 : q(1,1,1); 1,2 : q(1,1,2);

2,1 : q(1,2,1); 2,2 : q(1,2,2)} <= 3.

:- p(2), #sum{1 : q(2,1,1); 1 : q(2,1,2);

2 : q(2,2,1); 2 : q(2,2,2)} >= 2,

#sum{1,1 : q(2,1,1); 1,2 : q(2,1,2);

2,1 : q(2,2,1); 2,2 : q(2,2,2)} <= 3.

These rules form the basis for Example 2. �

An interpretation I is a set of atoms; atoms in I

are true, those not in I are false. In order to apply I

to ground aggregates, first consider a set S of ground

aggregate elements, and let I(S) be the following set of

tuples:

{〈t1, . . . , tm〉 | (t1, . . . , tm : ϕ) ∈ S ∧ I |= ϕ}.

essentially comprising the tuples of ground terms asso-

ciated with true conjunctions. The set I(S) is used to

obtain a multiset by collecting the first elements of all

tuples, on which the aggregation function is applied:

I(f(S)) := f([t1 | 〈t1, . . . , tm〉 ∈ I(S)]).

The natural interpretation of the comparator defines

satisfaction:

I |= f(S)� t if and only if I(f(S))� t holds.

Example 2 Let S1, S2 be the following sets of ground

aggregate elements from Example 1:

S1 := {1 : q(1, 1, 1); 1 : q(1, 1, 2);

2 : q(1, 2, 1); 2 : q(1, 2, 2)}
S2 := {1, 1 : q(1, 1, 1); 1, 2 : q(1, 1, 2);

2, 1 : q(1, 2, 1); 2, 2 : q(1, 2, 2)}

and let I be {q(1, 1, 1), q(1, 1, 2), q(1, 2, 1)}. Then I(S1)

is {〈1〉, 〈2〉} and I(S2) is {〈1, 1〉, 〈1, 2〉, 〈2, 1〉}. Hence,

I(#sum(S1)) = #sum([1, 2]) = 3, and therefore I satis-

fies #sum(S1) >= 2. On the other hand, I(#sum(S2)) =

#sum([1, 1, 2]) = 4, and #sum(S2) <= 3 is not satisfied

by I. �

3 Typical Use Cases

In many ASP programs, functional dependencies are

enforced by means of count aggregates, as in the fol-

lowing rule:

:- time(T), #count{A : do(A,T)} > 1.

The above rule in fact constrains relation do to satisfy

the functional dependency do[2]→ do[1], that is, there

must be a unique value of the first argument of relation

do for any value in the second argument of do. In this

case, the functional dependency stems from a planning

scenario where at most one action can be done at each

time step.

Similarly, aggregate functions are also commonly

used in ASP to constrain nondeterministic guesses, as

for example in the knapsack problem, where the total

weight of the selected items must not exceed a given

limit, which can be modeled by means of a rule of the

following form:

:- lim(L), #sum{W,O: obj(O,W,V), in(O)} > L.

where O is an object with weight W and value V. In this

case using W,O rather than just W guarantees that mul-

tiple objects with the same weight are accounted for

correctly.

The examples above use aggregation functions in

a rather simple form, that is, only in integrity con-

straints, which are rules with empty heads. However,

ASP does not restrict the use of aggregation functions

in rules with nonempty heads, which allows for repre-

senting problems that require inductive definitions in-

volving aggregates in rule bodies.

As an example, consider the company controls prob-

lem in the area of stock markets, in which the goal is to

determine pairs of companies x, y such that x controls y,

given information on stock possessions of companies. In

order to control a company y, a company x has to exert

control over more than 50% of the stock of y. Company

x can exert control over stock of y in two different ways:

x itself may possess a certain percentage of stock of y,

in this case we say that x directly exerts control over

the respective percentage of stock of y. Alternatively, if

x controls a company z, which exerts control over stock

of y, then we say that x indirectly exerts control over

the respective stock of y. Company x then exerts con-

trol over the sum of stock of y over which it directly

and indirectly exerts control.

Alternatively, we can define the control relation in-

ductively, by saying that company x controls y if the

sum of the shares in y possessed by x and by companies

controlled by x is more than 50%. The following rule

encodes this inductive definition:

controls(X,Y) :- company(X), company(Y),

#sum{S : owns(X,Y,S);

S,Z : controls(X,Z), owns(Z,Y,S)} > 50.

In this case, for each pair x, y of ground terms, the

aggregation is on the multiset obtained from the set

{〈s〉 | owns(x, y, s)} ∪
{〈s, z〉 | controls(x, z) ∧ owns(z, y, s)}

of tuples by projecting all elements but the first. In-

terestingly, the company controls problem, as stated

above, can be solved in polynomial time, but repre-

sented a challenge for ASP systems due to the presence

of recursive aggregates.

Inductive definitions involving aggregates can be

combined with other constructs of ASP, as for example

with choice rules, often used to define the search space

for nondeterministic guesses. Continuing with the com-

pany control problem, one may be interested in check-

ing whether the acquisition of some shares, available on

the stock exchange, may guarantee to exert control on

some target companies. In this case, if available pack-

ages of shares are represented by predicate onSale, the

following choice rule and inductive definition can be

used in combination with an integrity constraint to as-

sert the target controls:

{buy(Pkg)} :- onSale(Pkg,Comp,Share,Cost).

controls(X,Y) :- company(X), company(Y),

#sum{S : owns(X,Y,S);

S,Z : controls(X,Z), owns(Z,Y,S);

S,P : buy(P), onSale(P,Y,S,C)} > 50.

:- target(X,Y), not control(X,Y).

Note that in this case, for each pair x, y of ground terms,

the aggregation is on the multiset resulting from the set

{〈s〉 | owns(x, y, s)} ∪
{〈s, z〉 | controls(x, z) ∧ owns(z, y, s)} ∪
{〈s, p〉 | buy(p) ∧ ∃c onSale(p, y, s, c)},

where packages and companies are assumed to have dis-

joint identifiers. This problem belongs to the first level

of the polynomial hierarchy (it is NP-complete), and

its source of complexity is not the aggregation, but the

combination of choice rules and integrity constraints.

As a final use case, consider the generalized sub-

set sum problem [1,6,7,13], a prototypical problem for

the second level of the polynomial hierarchy, that is, a

ΣP
2 -complete problem. In the generalized subset sum

problem, two vectors u, v of integers, and an integer b

are given, and the task is to decide whether the formula

∃x∀y(ux+ vy 6= b)

is true, where x and y are vectors of binary variables of

the same length as u and v, respectively. For example,

for u = [1, 2], v = [2, 3], and b = 5, the task is to decide

whether the following formula is true:

∃x1x2∀y1y2(1 · x1 + 2 · x2 + 2 · y1 + 3 · y2 6= 5).

By combining several constructs and patterns provided

by ASP, such a complex computational problem can be

naturally encoded by the following rules:

{true(exists,X,C)} :- var(exists,X,C).

true(forall,Y,C) :- var(forall,Y,C),unequal.

:- not unequal.

unequal :- #sum{C,Q,X : true(Q,X,C)} != b.

First, a choice rule is used to encode the existential

guess, while the second and third rules are used to en-

code the universal quantification on variables Ys. Intu-

itively, any model I of the program is forced to con-

tain true(forall,y,c) for every var(forall,y,c) in

the input because of the integrity constraint; however,

the integrity constraint does not belong to the program

reduct, and therefore I is stable only if there is no as-

signment for variables Ys such that the aggregate in the

last rule becomes false. Indeed, for the following facts:

var(exists,1,1). var(exists,2,2).

var(forall,1,2). var(forall,2,3).

#const b = 5.

the program admits a unique stable model represent-

ing the only solution for u = [1, 2], v = [2, 3], and

b = 5, that is, x1 true and x2 false. In this case, the

aggregation is one of the source of complexities of the

above encoding, as in fact it does not belong to the class

of convex aggregates, the complexity boundary of ASP

programs under F-stable semantics [4].

4 History and Alternative Semantics

Logic programs had been endowed with aggregates for

quite some time in Datalog, a database query language.

In the late 1990ies these were also introduced to ASP,

first to enhance the languages of ASP systems. In

smodels [39], weight and cardinality constraints were

supported [36], while dlv [29,8] had a slightly more

general framework similar to the one presented in this

paper [20]. While smodels allowed for aggregates in

recursive definitions, dlv did not for some time.

It was soon realized that recursive definitions pose

some peculiar semantic issues. In order to overcome

these, initially two kinds of semantics were suggested:

one sometimes termed PSP (for Pelov [37], and Son and

Pontelli [40]) and another known as FLP (for Faber,

Leone, Pfeifer [19], and Ferraris [21]). The latter forms

the basis for the semantics mentioned here (F-stable,

for Ferraris), which deals with negated aggregates in

an arguably better way. These two groups of semantics

coincide on a significant number of programs, namely

all programs that do not contain aggregates occurring

in recursion, and programs that contain only aggregates

that are convex. Many aggregates are indeed convex, a

notable exception being #sum when the summation is

extended for both positive and negative numbers.

While the discussion on the appropriate semantics

for aggregates is still continuing (see for example [5,26,

38]), it appears as if there were consensus on programs

that contain only aggregates that are convex, and in

particular on programs that do not involve aggregates

in inductive definitions. The latter have a clear seman-

tics, and even most of the earliest semantics agree on

them. Moreover, most programs seen in practice at the

moment fall into this class. And for many semantics,

going beyond convex aggregates increases the compu-

tational complexity of many reasoning tasks in the gen-

eral case (if P 6= NP; see for example [10,11]). In this

respect, while convex is the complexity boundary of F-

stable models [4], this is not the case for PSP, for which

there are several non-convex aggregations that do not

increase the complexity of common reasoning tasks [9].

Complexity arguments also motivated the introduc-

tion of several techniques to normalize ASP programs

with aggregates since the beginning [39]. Specifically,

normalized ASP programs only contain monotone ag-

gregates, and therefore some rewritings are correct only

in the stratified case [30]. In fact, any polynomial, faith-

ful, and modular translation [28] to compile logic pro-

grams with non-convex aggregates into equivalent logic

programs that only comprise monotone aggregates

must introduce a different source of complexity. In a re-

cent polynomial, faithful, and modular translation im-

plemented in gringo [22], this source of complexity is

recursive disjunction in rule heads [6,7]. Other relevant

rewriting techniques in the literature aim to completely

eliminate aggregates [15,14], and proved to be quite ef-

ficient in practice; these rewritings produce aggregate-

free programs preserving F-stable models only in the

stratified case, or if recursion is limited to convex ag-

gregates, which is guaranteed to be the case for the

output of the latest versions of gringo.

More extensive discussions of the history of aggre-

gates in ASP can be found in the works of Pelov [37]

and Ferraris [21].

E. Fermé, J. Leite (eds.) JELIA 2014, Funchal, Madeira,
Portugal, September 24-26, 2014. Proceedings, Lecture
Notes in Computer Science, vol. 8761, pp. 166–180.
Springer (2014). DOI 10.1007/978-3-319-11558-0\ 12

15. Bomanson, J., Janhunen, T.: Normalizing cardi-
nality rules using merging and sorting construc-



6 Mario Alviano, Wolfgang Faber

tions. pp. 187–199. Springer (2013). DOI 10.1007/
978-3-642-40564-8\ 19. URL http://dx.doi.org/10.

1007/978-3-642-40564-8_19

16. Brewka, G., Eiter, T., Truszczynski, M.: Answer set pro-
gramming at a glance. Commun. ACM 54(12), 92–103
(2011). DOI 10.1145/2043174.2043195

17. Bruynooghe, M., Blockeel, H., Bogaerts, B., de Cat, B.,
Pooter, S.D., Jansen, J., Labarre, A., Ramon, J., De-
necker, M., Verwer, S.: Predicate logic as a modeling
language: modeling and solving some machine learning
and data mining problems with IDP3. TPLP 15(6),
783–817 (2015). DOI 10.1017/S147106841400009X. URL
https://doi.org/10.1017/S147106841400009X

18. Devriendt, J., Bogaerts, B., Bruynooghe, M., Denecker,
M.: On local domain symmetry for model expan-
sion. TPLP 16(5-6), 636–652 (2016). DOI 10.1017/
S1471068416000508. URL https://doi.org/10.1017/

S1471068416000508

19. Faber, W., Pfeifer, G., Leone, N.: Semantics and com-
plexity of recursive aggregates in answer set program-
ming. Artif. Intell. 175(1), 278–298 (2011). DOI
10.1016/j.artint.2010.04.002

20. Faber, W., Pfeifer, G., Leone, N., Dell’Armi, T., Ielpa,
G.: Design and implementation of aggregate functions in
the DLV system. TPLP 8(5-6), 545–580 (2008). DOI
10.1017/S1471068408003323

21. Ferraris, P.: Logic programs with propositional connec-
tives and aggregates. ACM Trans. Comput. Log. 12(4),
25 (2011). DOI 10.1145/1970398.1970401

22. Gebser, M., Kaminski, R., König, A., Schaub, T.: Ad-
vances in gringo series 3. In: J.P. Delgrande, W. Faber
(eds.) LPNMR 2011, Vancouver, Canada, May 16-19,
2011. Proceedings, Lecture Notes in Computer Science,
vol. 6645, pp. 345–351. Springer (2011). DOI 10.1007/
978-3-642-20895-9\ 39. URL http://dx.doi.org/10.

1007/978-3-642-20895-9_39

23. Gebser, M., Kaufmann, B., Schaub, T.: Conflict-driven
answer set solving: From theory to practice. Artif. Intell.
187, 52–89 (2012). DOI 10.1016/j.artint.2012.04.001

24. Gelfond, M., Lifschitz, V.: The stable model semantics
for logic programming. In: R.A. Kowalski, K.A. Bowen
(eds.) Logic Programming, Proceedings of the Fifth In-
ternational Conference and Symposium, Seattle, Wash-
ington, August 15-19, 1988 (2 Volumes), pp. 1070–1080.
MIT Press (1988)

25. Gelfond, M., Lifschitz, V.: Classical negation in logic pro-
grams and disjunctive databases. New Generation Com-
put. 9(3/4), 365–386 (1991). DOI 10.1007/BF03037169

26. Gelfond, M., Zhang, Y.: Vicious circle principle and
logic programs with aggregates. TPLP 14(4-5), 587–601
(2014). DOI 10.1017/S1471068414000222

27. Giunchiglia, E., Lierler, Y., Maratea, M.: Answer set pro-
gramming based on propositional satisfiability. J. Au-
tom. Reasoning 36(4), 345–377 (2006). DOI 10.1007/
s10817-006-9033-2

28. Janhunen, T.: Some (in)translatability results for nor-
mal logic programs and propositional theories. Journal
of Applied Non-Classical Logics 16(1-2), 35–86 (2006).
DOI 10.3166/jancl.16.35-86. URL http://dx.doi.org/

10.3166/jancl.16.35-86

29. Leone, N., Pfeifer, G., Faber, W., Eiter, T., Gottlob, G.,
Perri, S., Scarcello, F.: The DLV system for knowledge
representation and reasoning. ACM Trans. Comput. Log.
7(3), 499–562 (2006). DOI 10.1145/1149114.1149117.
URL http://doi.acm.org/10.1145/1149114.1149117

30. Liu, G., You, J.: Relating weight constraint and aggregate
programs: Semantics and representation. TPLP 13(1),
1–31 (2013). DOI 10.1017/S147106841100038X. URL
http://dx.doi.org/10.1017/S147106841100038X

31. Liu, L., Pontelli, E., Son, T.C., Truszczynski, M.: Logic
programs with abstract constraint atoms: The role of
computations. Artif. Intell. 174(3-4), 295–315 (2010).
DOI 10.1016/j.artint.2009.11.016

32. Liu, L., Truszczynski, M.: Properties and applications of
programs with monotone and convex constraints. J. Ar-
tif. Intell. Res. (JAIR) 27, 299–334 (2006). DOI 10.1613/
jair.2009. URL http://dx.doi.org/10.1613/jair.2009

33. Maratea, M., Pulina, L., Ricca, F.: A multi-engine ap-
proach to answer-set programming. TPLP 14(6), 841–
868 (2014). DOI 10.1017/S1471068413000094

